DYNARE SYNTAX

Michel Juillard

Paris–Jourdan Sciences Economiques
CEPREMAP
University Paris 8

Exemple

Cash in advance model from F. Schorfheide (2000) “Loss function-based evaluation of DSGE models” *Journal of Applied Econometrics*, *15*, 645-670.
See also J. Nason and T. Cogley (1994) “Testing the implications of long-run neutrality for monetary business cycle models” *Journal of Applied Econometrics*, *9*, S37-S70.

CIA model (I)

```
// declaration of endogenous variables
var gM P c W R k d n l y gA;

// declaration of exogenous shocks
varexo e_a e_m;

// declaration of parametrs
parameters alp bet gam mst rho psi del;

// setting values for parameters
alp = 0.33;
bet = 0.99;
gam = 0.003;
mst = 1.011;
rho = 0.7;
psi = 0.787;
del = 0.02;
```

CIA model (II)

```
// equations
model;
gA = gam+e_a;
gM = (1-rho)*log(mst) + rho*gM(-1)+e_m;
-P/(c(+1)*P(+1)*exp(gM))+bet*P(+1)
  *(alp*exp(-alp*gA(+1))*k^(alp-1)*n(+1)^(1-alp)+(1-del)*exp(-gA(+1)))
  /(c(+2)*P(+2)*exp(gM(+1)))=0;
W = l/n;
-(psi/(1-psi))*(c*P/(1-n))+l/n = 0;
R = P*(1-alp)*exp(-alp*gA)*k(-1)^alp*n^(-alp)/W;
1/(c*P)-bet*P*(1-alp)*exp(-alp*gA)*k(-1)^alp*n^(1-alp)/
  (exp(gM)*l*c(+1)*P(+1)) = 0;
c+k = exp(-alp*gA)*k(-1)^alp*n^(1-alp)+(1-del)*exp(-gA)*k(-1);
P*c = exp(gM);
exp(gM)-1+d = l;
y = k(-1)^alp*n^(1-alp)*exp(-alp*gA);
end;
```

CIA model (III)

```
// standard deviation of shocks
shocks;
var e_a; stderr 0.014;
var e_m; stderr 0.005;
end;
```

CIA model (IV)

```
// guess values for steady state computation
initval;
k = 6;
gM = log(mst);
P = 2.25;
c = 0.45;
W = 4;
R = 1.02;
d = 0.85;
n = 0.19;
l = 0.86;
y = 0.6;
gA = gam;
end;
```

CIA model (V)

```
// computes the steady state
steady;

// computes the eigenvalues
check;

// computes the solution
stoch_simul(order=1);
```

Syntax conventions

- comments are indicated by `//`
- exogenous shocks can only appear at the current period
- variable are set current in the period when they are decided upon
- stock variables must be on a *stock at the end of the period* basis